Combining Heuristics for Configuration Problems Using Answer Set Programming

Martin Gebser[1], Anna Ryabokon[2], and Gottfried Schenner[3]

[1] Aalto University, HIIT, Finland and University of Potsdam, Germany
martin.gebser@aalto.fi
[2] Alpen-Adria-Universität Klagenfurt, Austria
anna.ryabokon@aau.at
[3] Siemens AG Österreich, Vienna, Austria
gottfried.schenner@siemens.com

Abstract. This paper describes an abstract problem derived from a combination of Siemens product configuration problems encountered in practice. Often isolated parts of configuration problems can be solved by mapping them to well-studied problems for which efficient heuristics exist (graph coloring, bin-packing, etc.). Unfortunately, these heuristics may fail to work when applied to a problem that combines two or more subproblems. In the paper we show how to formulate a combined configuration problem in Answer Set Programming (ASP) and to solve it using heuristics à la hclasp. In addition, we present a novel method for heuristic generation based on a combination of greedy search with ASP that allows to improve the performance of an ASP solver.

Keywords: configuration problem, heuristics, answer set programming

1 Introduction

Configuration is a design activity aiming at creation of an artifact from given components such that a set of requirements reflecting individual needs of a customer and compatibility of the system's structures are satisfied. Configuration is a fully or partly automated approach supported by a knowledge-based information system called *configurator*. Originally configurators appeared because configurable products were expensive and very complex. They were developed by a significant number of highly qualified workers by order and single-copy. Emerging research on expert systems in the 1980s resulted in a number of approaches to knowledge-based configuration, such as McDermott's R1/XCON 'configurer' [13]. Since then many companies such as ConfigIt, Oracle, SAP, Siemens or Tacton have developed configurators for large complex systems reducing the production costs significantly. With the lapse of time the focus has been shifted more in the direction of mass customization. Currently configurators cover a wide range of customers and can be found in practically every price segment. One can configure a car, a computer, skis and even a forage for a dog.

Researchers in academia and industry have tried different approaches to configuration knowledge representation and reasoning, including production rules, constraints

languages, heuristic search, description logics, etc.; see [19, 17, 10] for surveys. Although constraint-based methods remain de facto standard, ASP has gained much attention over the last years because of its expressive high-level representation abilities. Normal rules as well as rules including weight and cardinality atoms were used in the first application of ASP to configuration problems [18]. Regarding knowledge representation, [21] suggests a high-level object-oriented modeling language and a web-based graphical user interface to simplify the modeling of requirements.

In [5] important aspects for formalizing and tackling real-world configuration scenarios with ASP are discussed. Recently a framework for describing object-oriented knowledge bases was presented in [15]. The authors suggested a general mapping from an object-oriented formalism to ASP for S'UPREME based configurators. S'UPREME is a configuration engine of Siemens AG, which is applied to configure complex large-scale technical systems such as railway safety systems within Siemens. In fact, more than 30 applications are based on this system [9].

As evaluation shows ASP is a compact and expressive method to capture configuration problems [10], i.e. it can represent configuration knowledge consisting of component types, associations, attributes, and additional constraints. The declarative semantics of ASP programs allows a knowledge engineer to freely choose the order in which rules are written in a program, i.e. the knowledge about types, attributes, etc. can be easily grouped in one place and modularized. Sound and complete solving algorithms allow to check a configuration model and support evolution tasks such as reconfiguration. However, empirical assessments indicate that ASP has limitations when applied to large-scale product configuration instances [1, 5]. The best results in terms of runtime and solution quality were achieved when domain-specific heuristics were used [20, 14].

In this paper we introduce a combined configuration problem that reflects typical requirements frequently occurring in practice of Siemens. The parts of this problem correspond (to some extent) to classical computer science problems for which there already exist some well-known heuristics and algorithms that can be applied to speed up computations and/or improve the quality of solutions.

As the main contribution, we present a novel approach on how problem-specific heuristics generated by a greedy solver can be incorporated in an ASP program to improve computation time (and obtain better solutions). The application of domain-specific knowledge formulated succinctly in an ASP heuristic language [8] allows for better solutions within a shorter solving time, but it strongly deteriorates the search when additional requirements (conflicting with the formulated heuristics) are included. On the other hand, the formulation of complex heuristics might be cumbersome using greedy methods. Therefore, we exploit a combination of greedy methods with ASP for the generation of heuristics and integrate them to accelerate an ASP solver. We evaluate the method on a set of instances derived from configuration scenarios encountered by us in practice and in general. Our evaluation shows that solutions for three sets of instances can be found an order of magnitude faster than compared to a plain ASP encoding.

In the following, Section 2 introduces a combined configuration problem (CCP) which is exemplified in Section 3. Its ASP encoding is shown in Section 4. Section 5 discusses heuristics for solving the CCP problem and we present our evaluation results in Section 6. Finally, in Section 7 we conclude and discuss future work.

2 Combined Configuration Problem

The Combined Configuration Problem (CCP) is an abstract problem derived from a combination of several problems encountered in Siemens practice (railway interlocking systems, automation systems, etc.). A CCP instance is defined by a directed acyclic graph (DAG). Each vertex of the DAG has a type and each type of the vertices has a particular size. In addition, each instance comprises two sets of vertices specifying two vertex-disjoint paths in the DAG. Furthermore, an instance contains a set of areas, sets of vertices defining possible border elements of each area and a maximal number of border elements per area. Finally, a number of available colors as well as a number of available bins and their capacity are given.

Given a CCP instance, the goal is to find a solution that satisfies a set of requirements. All system requirements are separated into the corresponding subproblems which must be solved together or in particular combinations:

- **P1 Coloring** *Every vertex must have exactly one color.*
- **P2 Bin-Packing** *For every color a Bin-Packing problem must be solved, where the same number of bins are available for each color. Every vertex must be assigned to exactly one bin of its color and for every bin, the sum of sizes must be smaller or equal to the bin capacity.*
- **P3 Disjoint Paths** *Vertices of different paths cannot be colored in the same color.*
- **P4 Matching** *Each border element must be assigned to exactly one area such that the number of selected border elements of an area does not exceed the maximal number of border elements and all selected border elements of an area have the same color.*
- **P5 Connectedness** *Two vertices with the same color must be connected via a path that contains only vertices of that color.*

Origin of the problem The considered CCP originates in the *railway domain*. The given DAG represents a track layout of a railway line. A coloring **P1** can then be thought as an assignment of resources (e.g. computers) to the elements of the railway line. In real-world scenarios different infrastructure elements may require different amounts of a resource that is summarized in **P2**. This may be hardware requirements (e.g. a signal requiring a certain number of hardware parts) or software requirements (e.g. an infrastructural element requiring a specific processing time). The requirements of **P1** and **P2** are frequently used in configuration problems during an assignment of entities of one type to entities of another type [12, 5]. The constraint of **P3** increases availability, i.e. in case one resource fails it should still be possible to get from a source vertex (no incoming edges) of the DAG to a target vertex (no outgoing edges) of the DAG. In the general version of this problem one has to find n paths that maximize availability. The CCP uses the simplified problem where 2 vertex-disjoint paths are given. **P4** stems from detecting which elements of the graph are occupied. The border elements function as detectors for an object leaving or entering an area. The PUP problem [2, 1] is a more elaborate version of this problem. **P5** arises in different scenarios. For example, if communication between elements controlled by different resources is more costly, then neighboring elements should be assigned to the same resource whenever possible.

3 Example

Fig. 1 shows a sample input CCP graph. In this section we illustrate how particular requirements can influence a solution. Namely, we add the constraints of each subproblem one by one. If only **P1** is active, any graph corresponds to a trivial solution of **P1** where all vertices are colored white.



Fig. 1. Input CCP graph and a trivial solution of Coloring (**P1**)

Let us consider the input graph as a Bin-Packing problem instance with four colors and three bins per color of a capacity equal to five. The vertices of type b, e, s and p have the sizes 1, 2, 3 and 4 respectively. A sample solution of Coloring and Bin-Packing (**P1-P2**) is presented in Fig. 2 and Fig. 3.

For instance, when activating the Disjoint Paths constraint (**P3**), two vertex-disjoint paths $path1 = \{b1, s1, p1, b2, p2, b3, p3, s2, b4\}$ as well as $path2 = \{b7, s3, p4, b8, p5, b9, p6, s4, b10\}$ may be declared. Consequently, in this case the solution shown in Fig. 2 violates the constraint and must be modified as displayed in Fig. 4, where the vertices of different paths are colored with different colors ($path1$ with dark grey and grey whereas white and light grey are used for $path2$).



Fig. 2. Used colors in a solution of the Coloring and Bin-Packing problems (**P1**-**P2**)

Fig. 5 shows a Matching example (**P4**). There are seven areas in the matching input graph, each corresponding to a subgraph surrounded with border elements (Fig. 1). For example, area $a1$ represents the subgraph $\{b1, s1, p1, b2, b5\}$ and area $a2$ the subgraph $\{b5, e1, b6\}$. The corresponding border elements are $\{b1, b2, b5\}$ and $\{b5, b6\}$ (Fig. 5).

Assume that an area can have at most 2 border elements assigned to it. In the resulting matching (Fig. 5) $b1$, $b2$ are assigned to $a1$ whereas $b5$, $b6$ are assigned to $a2$. Note that the sample selected matching shown in Fig. 5 is not valid with the coloring



Fig. 3. Used bins in a solution of the Coloring and Bin-Packing problems (**P1**-**P2**)



Fig. 4. Solution of the Coloring, Bin-Packing and Disjoint Paths problems (**P1**-**P3**)



Fig. 5. A sample input and solution graphs for **P4**. The selected edges of the input graph are highlighted with solid lines.



Fig. 6. A valid solution for **P1**-**P5**

presented previously, because, for example, $b5$ and $b6$ are assigned to the same area $a2$ although they are colored differently. In addition, the coloring solution shown in Fig. 4 violates the Connectedness constraint (**P5**). Therefore, the previous solutions must be updated to take the additional requirements into account. Fig. 6 shows a valid coloring of the given graph that satisfies all problem conditions (**P1**-**P5**).

4 ASP encoding of the Combined Configuration Problem

A CCP instance is defined using the following atoms. An edge between two vertices in the DAG is defined by $edge(Vertex1, Vertex2)$. For each vertex, $type(Vertex, Type)$ and $size(Vertex, Size)$ are declared. $pathN(Vertex)$ expresses that a vertex belongs to a particular path. In addition, each border element must be connected to one of the

possible areas given by $edge\_matching(Area, Vertex)$ whereas each area can control at most $maxborder(C)$ border elements. The number of colors and bins are defined using $nrofcolors(Color)$ and $nrofbins(Bin)$. Finally, the capacity of a bin is fixed by $maxbinsize(Capacity)$.

Our ASP encoding for the CCP is shown in Listing 1. Line 1-5 implements Coloring (**P1**), assigning colors to vertices. The atoms $vertex\_color(Vertex, Color)$ and $usedcolor(Color)$ express that a $Vertex$ is connected to a $Color$, i.e. used in a solution via $usedcolor(Color)$. An assignment of a $Vertex$ to a Bin, i.e. Bin-Packing problem (**P2**), is accomplished using Line 6-10, where the atoms $vertex\_bin(Vertex, Bin)$ and $usedbin(Bin)$ represent a solution. Further, the atoms $bin(Color, Bin, Vertex)$ represent a combined solution for **P1** and **P2**. The Disjoint paths constraint (**P3**) is stated in Line 11. In accordance with Matching (**P4**), i.e. Line 12-17, one has to find a matching between areas and border elements using $edge\_matching\_selected(Area, Vertex)$ atoms. Finally, the Connectedness requirement (**P5**) is ensured in line 18-24.

```
1  vertex(V):-type(V,_). vertex(V):-size(V,_). % P1
2  vertex(V):-edge(V,_). vertex(V):-edge(_,V).
3  color(1..MaxC):-nrofcolors(MaxC).
4  1{vertex_color(V,C):color(C)}1:-vertex(V).
5  usedcolor(C):-vertex_color(V,C).

6  1{vertex_bin(V,B):B=1..K}1 :- vertex(V), nrofbins(K). % P2
7  bin(C,B,V):-vertex_color(V,C),vertex_bin(V,B).
8  :-color(C),nrofbins(K),maxbinsize(MaxS), B=1..K,
9    MaxS+1 #sum{S,V:bin(C,B,V),size(V,S)}.
10 usedbin(B):-bin(C,B,V).

11 :-path1(V1),path2(V2),vertex_color(V1,C),vertex_color(V2,C).%P3

12 area(A):-edge_matching(A,B). % P4
13 borderelement(B):-edge_matching(A,B).
14 1{edge_matching_selected(A,B):edge_matching(A,B)}1 :-
       borderelement(B).
15 :-area(A),maxborder(MaxB),MaxB+1{edge_matching_selected(A,B)}.
16 edge_matching_color(A,C):-edge_matching_selected(A,B),
       vertex_color(B,C).
17 :-area(A), 2{edge_matching_color(A,C)}.

18 e(X,Y):-edge(X,Y). e(X,Y):-edge(Y,X). % P5
19 pred(V1,V2):-vertex(V1;V2),V1 < V2,V <= V1:vertex(V),V<V2.
20 first(C,V2):-color(C), pred(V1,V2),
21           not vertex_color(V1,C),first(C,V1):pred(V,V1).
22 reach_col(C,V1):-color(C),vertex(V1),first(C,V1):pred(V,V1).
23 reach_col(C,V2):-reach_col(C,V1),e(V1,V2),vertex_color(V1,C).
24 :-vertex_color(V,C),not reach_col(C,V).
```

Listing 1. ASP encoding for the Combined Configuration Problem

5 Combining Heuristics for Configuration Problems

To formulate a heuristic within ASP we use the declarative heuristic framework developed by Gebser et al. [8]. In this formalism the heuristics are expressed using atoms $\_heuristic(a, m, v, p)$, where a denotes an atom for which a heuristic value is defined, m is one of four modifiers (init, factor, level and sign), and v, p are integers denoting a value and a priority, respectively, of the definition. A number of shortcuts are available, e.g. $\_heuristic(a, v, l)$, where a is an atom, v is its truth value and l is a level. The heuristic atoms modify the behavior of the VSIDS heuristic [11]. Thus, if a $\_heuristic$ atom is true in some interpretation, then the corresponding atom a might be preferred by the ASP solver at the next decision point. For instance, given the choice rule $1\{vertex\_color(V, C) : color(C)\}1$:- $vertex(V)$. and adding only the atom $\_heuristic(vertex\_color(\text{'b1'}, 1), true, 1))$ to a program, the solver prefers the atom $vertex\_color(\text{'b1'}, 1)$ over all other atoms $vertex\_color(\text{'b1'}, X)$ for $X \neq 1$. If several atoms $vertex\_color/2$ are provided, the atom with the higher level l is preferred.

There are different ways to incorporate heuristics in a program. The standard approach [8] requires an implementation of a heuristic at hand using a pure ASP encoding, whereas the idea of our method is to delegate the (expensive) generation of a heuristic to an external tool and then to extend the program with generated heuristic atoms to accelerate the ASP search. Below we exemplify how both approaches can be applied.

5.1 Standard generation of heuristics in ASP

Several heuristics can be used for the problems that compose the CCP, e.g. for the coloring of vertices (**P1**) we seek to use as few colors as possible by the following rule:

```
1 _heuristic(vertex_color(V,C),true,MC-C) :- vertex(V), color(C),
      nrofcolors(MC).
```

Listing 2. Heuristic for an assignment of colors to vertices

Additionally, we can apply well-known Bin-Packing heuristics for the placement of colored vertices into the bins of specified capacity (**P2**). The Bin-Packing problem is known to be an NP-hard combinatorial problem. However, there is a number of approximation algorithms (construction heuristics) that allow efficient computation of good approximations of a solution [6], e.g. Best/First/Next-Fit heuristics. They can, of course, be used as heuristics for the CCP. As shown in Listing 3, given a (decreasing) order of vertices using $order(V, O)$ atoms, we can force the solver to place vertex V_i into the lowest-indexed bin for which the size of already placed vertices does not exceed the capacity, i.e. in a first-fit bin:

```
1 binDomain(1..NB) :- nrofbins(NB). offset(NB+1) :- nrofbins(NB).
2 _heuristic(vertex_bin(V,B),true,M+O*NB-B) :- binDomain(B),
      nrofbins(NB), order(V,O), offset(M).
```

Listing 3. First-Fit heuristic for an assignment of vertices to bins

The heuristic never uses a new bin until all the non-empty bins are full and it can be expressed by rules that generate always a higher level for the bins with smaller number. It is also possible (with an intense effort) to express other heuristics for **P1**-**P5** that guide the search appropriately and allow to speed up the computation of solutions if we solve these problems separately. However, as our experiments show, the inclusion of heuristics for different problem at the same time might drastically deteriorate the performance for real-world CCP instances.

5.2 Greedy Search

From our observations in the context of product configuration, it is relatively easy to devise a greedy algorithm to solve a part of a configuration problem. This is often the case in practice, because products are typically designed to be easily configurable. The hard configuration instances usually occur when new constraints arise due to the combination of existing products and technologies.

Algorithm 1: GreedyMatching

Input: A bipartite graph $G_A = (BE, A, E)$, where BE is a set of border elements, A is a set of areas and $E \subseteq BE \times A$ is a set of edges
Output: A matching set M

1 $M \leftarrow \emptyset$;
2 **foreach** $v \in BE$ **do**
  // Select areas with the minimum number of matched elements
3 $A' \leftarrow \arg\min_{a \in A} |\{v' \mid v' \in BE, (v', a) \in M, (v, a) \in E\}|$;
4 $a \leftarrow \mathtt{pop}(A')$;
5 $M \leftarrow M \cup \{(v, a)\}$;
6 **return** M;

The same can be said for the CCP problem. Whereas it is easy to develop greedy search algorithms for the individual subproblems, it becomes increasingly difficult to come up with an algorithm that solves the combined problem. Algorithm 1 shows a greedy method that solves the Matching problem of the CCP (**P4**). For every vertex v it finds a related area a with the fewest assigned vertices so far and matches v with a. The algorithm assumes that all border elements are colored with one color, as it trivially satisfies the coloring requirement of the matching problem. Algorithm 2 shows a greedy approach to solving the CCP wrt. Coloring, Bin-Packing and Connectedness (**P1**, **P2** and **P5**). Every call to `pop` returns and removes the first element v of the set V and all corresponding edges. Then, the vertex v is assigned a color and is put into a bin according to some heuristic Bin-Packing algorithm. For instance, one can use classic heuristics as First-Fit or Best-Fit [6]. Our implementation of `assignVertexToBin` puts vertices of only one color into a bin. If the number of bins K is not enough to pack a vertex, then the set of bins B is not modified and the vertex is ignored. In case the vertex was placed into a bin, Algorithm 2 retrieves and removes from G all vertices

Algorithm 2: GreedyColoringBinPackingConnectedness

Input: A graph $G = (V, E)$, a maximum number of bins K for each color and a bin capacity C
Output: A set B that comprises all bins of a solution

```
1  B ← ∅; color ← 1; Q ← ∅;
2  while V ≠ ∅ do
3      q ← pop(V); Q ← {q};
4      while Q ≠ ∅ do
5          v ← pop(Q);
6          labelVertexWithColor(v, color);
7          B ← assignVertexToBin(B, v, C, K);      // v is ignored, if it does not fit
8          if ∃b ∈ B (v ∈ b) then
9              V ← V \ {v};
10             Q ← Q ∪ popNeigbours(v, G);
11     color ← color + 1;
12 return B;
```

adjacent to v. The loop continues until all vertices that can be reached from v are colored and assigned to some bin. Finally, the number of colors is increased and the algorithm colors and removes another subgraph of G until no vertices in G are left.

Suppose one wants to combine these two algorithms. One strategy would be to run greedy Matching and then solve the Bin-Packing problem taking matchings into account. Thus, a combined algorithm first calls Algorithm 1 and gets a set of matchings $M = \{(v_1, a_1), \ldots, (v_n, a_m)\}$. Then, for each vertex v_i of the input graph G the algorithm (i) assigns a new color to v_i, if v_i has no assigned color, and (ii) puts v_i into a bin, as in Algorithm 2. In case v_i is a border element, the combined algorithm retrieves an area a_j that matches v_i in M and colors all vertices of this area in the same color as v_i.

The combined algorithm might violate the Connectedness property, because it colors all border vertices assigned to an area with the same color. However, these vertices are not necessarily connected. That is, there might be a solution with a different matching, but the greedy algorithm tests only one of all possible matchings. Moreover, there is no obvious way how to create an algorithm solving all 3 problems efficiently. This is a clear disadvantage of using ad-hoc algorithms in contrast to the usage a logic-based formalism like ASP, where the addition of constraints is just a matter of adding some rules to an encoding. On the other hand, domain-specific algorithms are typically faster and scale better than ASP-based or SAT-based approaches that cannot be used for very large instances. For instance, the memory demand of the greedy Algorithm 2 is almost independent of graph size.

5.3 Combining Greedy Search and ASP

One way to let a complete ASP solver and a greedy search algorithm benefit from each other is to use the greedy algorithm to compute upper bounds for the problem to solve. The tighter upper bound usually means smaller grounding size and shorter solving time

Algorithm 3: Greedy & ASP

Input: A problem P, an ASP program Π solving the problem P
Output: A solution S

```
1 GreedySolution ← solveGreedy(P);
2 H ← generateHeuristic(GreedySolution);
3 return solveWithASP(Π, H);
```

because the greedy solver being domain-specific usually outperforms ASP for the relaxed version of the problem. For instance, running the greedy algorithm for the Bin-Packing problem and Matching problem gives upper bounds for the maximal number of colors, i.e. number of different Bin-Packing problems to solve. The same applies to the Matching problem. This kind of application of greedy algorithms has a long tradition in branch and bound search algorithms, where greedy algorithms are used to compute the upper bound of a problem. For an example see [22], where a greedy coloring algorithm is used to find an upper bound for the clique size in a graph in order to compute maximum cliques. In this paper we investigate a novel way to combine greedy algorithms and ASP (Algortihm 3). Given all required inputs, first, a greedy algorithm is used to solve the Matching and Bin-Packing problems. The greedy algorithm typically solves a relaxed version of the problem, therefore, the solution found by the greedy algorithm may not be a consistent solution for ASP. This solution is converted into a heuristic for an ASP solver by giving the atoms of the solution a higher heuristic value.

As an example for solving the complete CCP problem, we can, first, find an unconnected solution for the combination of Coloring, Bin-Packing, Disjoint paths and Matching problems (**P1**-**P4**), and then, use the ASP solver to fix the Connectedness property (**P5**). The idea of combining local search with a complete solver is also found in large neighborhood search [4].

6 Experimental results

Experiment1 In our evaluation we compared a plain ASP encoding of the CCP with an ASP encoding extended with domain-specific knowledge. The Bin-Packing problem (**P2**) of the CCP corresponds to the classic Bin-Packing problem and the same heuristics can be applied. We implemented several Bin-Packing heuristics such as First/Best/Next-Fit (Decreasing) heuristics using ASP as shown in Section 5.1. For the evaluation we took 37 publicly available Bin-Packing problem instances[1], for which the optimal number of bins $optnrofbins$ is known, and translated them to CCP instances. The biggest instance of the set includes 500 vertices and 736 bins of the capacity 100. In the experiment, the maximal number of colors was set to 1 and the maximal number of bins was set to $2 \cdot optnrofbins$. All instances were solved by both approaches[2]. For a plain ASP encoding the solver required at most 27 seconds to find a solution whereas for

[1] The instances were taken from: http://www.wiwi.uni-jena.de/Entscheidung/binpp/index.htm.

[2] The evaluation was performed using clingo version 4.3.0 from the Potassco ASP collection [7] on a system with Intel i7-3030K CPU (3.20 GHz) and 64 GB of RAM, running Ubuntu 11.10.

the heuristic ASP program solving took at most 6 seconds, which is 4.5 times faster. The best results for the heuristic approach were obtained using the First-Fit heuristic with the decreasing order of vertices. Corresponding solutions utilized less bins then the ones obtained with the plain ASP program. Moreover, using First-Fit heuristic, for 23 from 37 instances a solution with optimal number of bins was found and for 13 other instances at most 4 bins more were required. The plain ASP encoding resulted in solutions that used on average 4 bins more than corresponding solutions of the heuristic approach. Only for 1 instance the heuristic program generated a worse solution than the plain ASP encoding.

Experiment2 In the next experiment we tested the same Bin-Packing heuristics implemented in ASP for the combined CCP, i.e. when all subproblems **P1**-**P5** are active, on 100 real-world test instances of moderate size (maximally 500 vertices in an input). The instances in this experiment were derived from a number of real-world configurations. Neither the plain program nor the heuristic programs were able to improve runtime/quality of solutions. Moreover, our greedy method described in Section 5.2 also failed to find a connected solution, i.e. when **P5** is active. For this reason, we investigated the combined approach (Greedy & ASP) described in Section 5.3. This approach uses the greedy method to generate a partial solution ignoring the Connectedness constraint and provides this solution as $\_heuristic$ atoms to the ASP solver. Our experiments show (see Fig. 7a) that the combined approach can solve all 100 benchmarks from the mentioned set, whereas the plain encoding presented in Section 4 solves only 54 instances (the time frame was set to 900 seconds in this and the next experiment). Moreover, for those instances which were solved using both approaches, the quality of solutions measured in terms of used bins and colors was the same. However, the runtime of the combined approach was 18 times faster on average and required at most 24 seconds instead of 848 seconds needed for the plain ASP encoding.

Experiment3 In addition, we tested more complex real-world instances (maximally 1004 vertices in an input)[3] which we have also submitted to the ASP competition 2015. Similarly to *Experiment2* we compared the plain ASP encoding from Section 4 to the combined approach in Section 5.3. Again, regarding the quality of solutions, both approaches are comparable, i.e. they use on average the same number of colors and bins, with the combined approach having a slight edge. Generally, from 48 instances considered in this experiment, 36/38 instances were solved using the plain/combined encoding, respectively. On average/maximally the plain encoding needed 69/887 seconds to find a solution whereas the combined method took 14/196 seconds, respectively, which is about 5 times faster. Fig. 7b shows the influence of heuristics on the performance for the instances from *Experiment3* that were solved by both approaches within 900 seconds. Although the grounding time is not presented for both experiments, we note that it requires about 10 seconds using both approaches for the biggest instance when all subproblems **P1**-**P5** are active.

[3] The instances are available at: http://isbi.aau.at/hint/problems



(a) *Experiment2* (b) *Experiment3*

Fig. 7. Evaluation results using Plain ASP and Greedy & ASP

7 Discussion

Choosing the right domain-specific heuristics for simple backtrack-based solvers is essential for finding a solution at all, especially for large and/or complex problems. The role of domain-specific heuristics in a conflict-driven nogood learning ASP solver seems to be less important when it comes to solving time. Here the size of the grounding and finding the right encoding is often the limiting factor. Nevertheless, *domain-specific heuristics are very important* to control the order in which answer sets are found and are an alternative to optimization statements. The latter hinder the computation of solutions for many configuration problem instances in a time which is reasonable for the application domain [1, 5]. As we have shown, domain-specific heuristics also provide a mechanism to combine greedy algorithms with ASP solvers, which opens up the possibility to use ASP in a meta-heuristic setting. However, the possible applications go beyond this. The same approach could be used to repair an infeasible assignment using an ASP solver. This is currently a field of active research for us and has applications in the context of product reconfiguration. Reconfiguration occurs when a configuration problem is not solved from scratch, but some parts of an existing configuration have to be taken into account.

An open question is how to combine heuristics for different subproblems in a modular manner without the adaptation of every domain-specific heuristic. Here approaches like search combinators [16] from the constraint programming community might be useful. Another interesting topic for future research would be how to learn heuristics from an ASP solver, i.e. to investigate the variable/value order chosen by an ASP solver for medium size problem instances and use them as heuristics in a backtrack solver for larger instances that are out of scope of an ASP solver due to the grounding size. Some aspects of this topic were discussed in [3]. Moreover, it is worthwhile to investigate how our method can be generalized to other application domains and whether we will be generally able to gain better performance if more heuristics are combined.

Acknowledgments
This work was funded by COIN and AoF under grant 251170 as well as by FFG under grant 840242. The authors would like to thank all anonymous reviewers for their comments and Konstantin Schekotihin for helpful discussions on the subject of this paper.

References

1. Aschinger, M., Drescher, C., Friedrich, G., Gottlob, G., Jeavons, P., Ryabokon, A., Thorstensen, E.: Optimization Methods for the Partner Units Problem. In: Proceedings of CPAIOR. pp. 4–19 (2011)
2. Aschinger, M., Drescher, C., Gottlob, G., Jeavons, P., Thorstensen, E.: Tackling the Partner Units Configuration Problem. In: Proceedings of IJCAI. pp. 497–503 (2011)
3. Balduccini, M.: Learning and using domain-specific heuristics in ASP solvers. AI Communications 24(2), 147–164 (2011)
4. Cipriano, R., Di Gaspero, L., Dovier, A.: A hybrid solver for large neighborhood search: Mixing gecode and easylocal++. In: Hybrid metaheuristics, pp. 141–155 (2009)
5. Friedrich, G., Ryabokon, A., Falkner, A.A., Haselböck, A., Schenner, G., Schreiner, H.: (Re) configuration based on model generation. In: LoCoCo workshop. pp. 26–35 (2011)
6. Garey, M.R., Johnson, D.S.: Computers and Intractability: A Guide to the Theory of NP-Completeness. W. H. Freeman (1979)
7. Gebser, M., Kaminski, R., Kaufmann, B., Schaub, T.: Answer Set Solving in Practice. Morgan & Claypool Publishers (2012)
8. Gebser, M., Kaufmann, B., Romero, J., Otero, R., Schaub, T., Wanko, P.: Domain-Specific Heuristics in Answer Set Programming. In: Proceedings of AAAI (2013)
9. Haselböck, A., Schenner, G.: S'UPREME. Knowledge-Based Configuration: From Research to Business Cases. pp. 263–269 (2014)
10. Hotz, L., Felfernig, A., Stumptner, M., Ryabokon, A., Bagley, C., Wolter, K.: Configuration Knowledge Representation and Reasoning. Knowledge-Based Configuration: From Research to Business Cases. pp. 41–72 (2014)
11. Madigan, C., Malik, S., Moskewicz, M., Zhang, L., Zhao, Y.: Chaff: Engineering an efficient SAT solver. In: Proceedings of DAC (2001)
12. Mayer, W., Bettex, M., Stumptner, M., Falkner, A.: On solving complex rack configuration problems using CSP methods. Proceedings of the IJCAI Workshop on Configuration (2009)
13. McDermott, J.: R1: A rule-based configurer of computer systems. Artificial Intelligence 19(1), 39–88 (1982)
14. Ryabokon, A., Friedrich, G., Falkner, A.A.: Conflict-Based Program Rewriting for Solving Configuration Problems. In: Proceedings of LPNMR. pp. 465–478 (2013)
15. Schenner, G., Falkner, A., Ryabokon, A., Friedrich, G.: Solving Object-oriented Configuration Scenarios with ASP. In: Proceedings of the Configuration Workshop. pp. 55–62 (2013)
16. Schrijvers, T., Tack, G., Wuille, P., Samulowitz, H., Stuckey, P.J.: Search combinators. Constraints 18(2), 269–305 (2013)
17. Sinz, C., Haag, A.: Configuration. IEEE Intelligent Systems 22(1), 78–90 (2007)
18. Soininen, T., Niemelä, I., Tiihonen, J., Sulonen, R.: Representing configuration knowledge with weight constraint rules. In: Proceedings of the Workshop on ASP. pp. 195–201 (2001)
19. Stumptner, M.: An overview of knowledge-based configuration. AI Communications 10(2), 111–125 (1997)
20. Teppan, E.C., Friedrich, G., Falkner, A.A.: QuickPup: A Heuristic Backtracking Algorithm for the Partner Units Configuration Problem. In: Proceedings of IAAI. pp. 2329–2334 (2012)
21. Tiihonen, J., Heiskala, M., Anderson, A., Soininen, T.: WeCoTin - A practical logic-based sales configurator. AI Communications 26(1), 99–131 (2013)
22. Tomita, E., Kameda, T.: An efficient branch-and-bound algorithm for finding a maximum clique with computational experiments. J Global Optim 37(1), 95–111 (2007)